
    Triage

A lightweight Elixir library for enhanced handling of results ({:ok, _} / :ok / {:error, _} / :error) with context wrapping, logging, and user message generation.
Features
This package provides three levels of working with errors which are all usable independently, but which all complement each other.
	Context Wrapping: Add meaningful context to errors as they bubble up through your application
	Result Logging: Log errors (and optionally successes) with file/line information
	User-friendly errors: Be able to collapse errors into a single user error message
	Error enumeration: functions like map_if, find_value, and all help deal with enumerations over data where each iteration may succeed or fail.
	Error control flow: ok_then and error_then functions help control and transform results

Design goals:
	Standard results (:ok, :error, {:ok, term()}, {:error, term()} with only one value in tuples)
	Avoid macros for easy pick-up-and-use throughout a codebase (i.e. no need for require)
	Variety of small tools which work well together (like UNIX commands)

See the Philosophy section of the docs for more details.
Examples
Contexts
When an error is returned (e.g. in a tuple, as opposed to being raised), often that error can be passed up a stack and it becomes unclear where the error came from. Triage offers a wrap_context function to attach a context string and/or metadata to errors via a WrappedError exception struct.
defmodule MyApp.OrderProcessor do
  def process_payment(order) do
    with {:ok, payment_method} <- fetch_payment_method(order),
         {:ok, charge} <- charge_payment(payment_method, order.amount) do
      {:ok, charge}
    end
    |> Triage.wrap_context("process payment", %{order_id: order.id, order_amount: order.amount})
  end
  # ...
end

defmodule MyApp.OrderService do
  def complete_order(order_id) do
    fetch_order(order_id)
    |> MyApp.OrderProcessor.process_payment()
    |> Triage.wrap_context("complete order")
  end
  # ...
end
But an error wrapped with a context isn't so useful by itself.  Your code can look at the WrappedError if you'd like, but it can be most useful with the output tools below.
(Also, make sure to see the Contexts section of the docs for more information)
Output
Error results that you get back can be a mess. Often when you get an error tuple it comes back from a tree of nested calls and the reason value could be of many types (string, atom, etc...).  So it's useful to have tools which let you not worry about it so much. Below is an example of using Triage.log to output logs:
def show(conn, %{"order_id" => order_id}) do
  order_id = String.to_integer(order_id)

  MyApp.complete_order(order_id)
  |> Triage.log()
  # ...
By default Triage.log will only output error cases (pass in :all to log :ok results as well), so if this case is important we can have a log of how it went wrong. Also note that any metadata given to log is also assigned to the Logger metadata in addition to being outputted (helpful for filtering logs).
The output can be as simple as this in the case of an atom given as the error reason:
[RESULT] lib/my_app/order_controller.ex:41: {:error, :order_was_invalid}
But if Triage.wrap_context is used, we can get even more details out:
[error] [RESULT] lib/my_app/order_service.ex:15: {:error, :payment_declined}
  [CONTEXT] lib/my_app/order_service.ex:15: complete order
  [CONTEXT] lib/my_app/order_processor.ex:8: process payment | %{order_id: 12345, amount: 99.99}
Note that if you'd prefer to output JSON logs, there is some information in the docs
Additionally, the Triage.user_message function will extract a message from the error if possible.  If not possible, the user will be given a generic error with a randomly generated short code which can be matched to a log entry with details about the error.
def show(conn, %{"order_id" => order_id}) do
  order_id = String.to_integer(order_id)

  MyApp.complete_order(order_id)
  |> case do
    {:ok, value} ->
      # ...

    {:error, _} = error ->
      conn
      |> put_status(400)
      |> json(%{error: Triage.user_message(error)})
  end
  # ...
The user_message function even supports a default implementation to provide Ecto.Changeset errors, so if it gets a changeset value in an error then users will get a reasonable value such as "age: must be greater than 18; email: can't be blank".
See the Outputs section of the docs for more information.
Enumeration
triage has a set of functions to help when you have a series of step which might succeed or fail.  As an example, you may want to build up a list, but return an error if anything fails.
  defp validate_each_metric(metrics, query) do
    Enum.reduce_while(metrics, {:ok, []}, fn metric, {:ok, acc} ->
      case validate_metric(metric, query) do
        {:ok, metric} -> {:cont, {:ok, acc ++ [metric]}}
        {:error, reason} -> {:halt, {:error, reason}}
      end
    end)
  end
The Triage.map_if function is one tool available:
  defp validate_each_metric(metrics, query) do
    # Returns {:ok, [...]} where the original returned just [...]
    Triage.map_if(metrics, & validate_metric(&1, query))
  end
For more functions and examples, see the Enumerating Errors section of the docs.
Control Flow
triage's two control flow tools (ok_then and error_then) can both be shown via an HTTP request example:
fetch_bill(bill_id)
|> Triage.ok_then(& HTTPoison.get(&1.pdf_url))
|> Triage.ok_then(fn
  %HTTPoison.Response{status_code: 200, body: body} ->
    body

  %HTTPoison.Response{status_code: 404, body: body} ->
    {:error, "Server result not found"}
end)
|> Triage.error_then(fn
    %HTTPoison.Error{reason: :nxdomain} ->
      "Server domain not found"

    %HTTPoison.Error{reason: :econnrefused} ->
      "Server connection refused"

    %HTTPoison.Error{reason: reason} ->
      "Unexpected error connecting to server: #{inspect(reason)}"
end)
The Triage.ok_then function works on :ok results, ignoring errors.  Values that are returned from the callback are automatically wrapped in an {:ok, _} tuple, though any :error or {:error, term()} returned will be returned as an error.
The Triage.error_then function is the opposite: working on :error reasons and returning new reasons to be wrapped in an {:error, _} tuple.  If an :ok or {:ok, _} result is returned, then the error is ignored and Triage.error_then will return that success.
Additionally there are tap_ok and tap_error function which allow you to execute side-effects (see also Elixir's Kernel.then/2 and Kernel.tap/2 functions which are analogous to the above).
Make sure to see the Control Flow section of the docs for more information.
Also, many people wonder why they shouldn't just use with instead of ok_then / error_then.  There is a section in the docs for that too!
Installation
Add triage to your list of dependencies in mix.exs:
def deps do
  [
    {:triage, "~> 0.5.0"}
  ]
end
For various reasons, triage requires at least version 1.15 of Elixir.
Usage
See the docs for detailed information about the different tools available.
Development
Run tests:
Run tests in watch mode (uses mix_test_interactive:
mix test.interactive

Or just:
mix test

License
Copyright (c) 2025
This work is free. You can redistribute it and/or modify it under the
terms of the MIT License. See the LICENSE file for more details.


  

    Changelog

All notable changes to this project will be documented in this file.
The format is based on Keep a Changelog,
and this project adheres to Semantic Versioning.
[0.7.1] - 2025-12-02
Fixed
	Documentation for Claude skill file

[0.7.0] - 2025-11-25
Fixed
	Support returning 3+ :ok/:error tuples from run/ok_then/error_then function callbacks

[0.6.0] - 2025-11-23
Added
	tap_ok and tap_error functions

[0.5.0] - 2025-11-21
Changed
	Renamed then!/1 and then/1 to run!/1 and run/1
	Renamed then!/2 and then/2 to ok_then!/2 and ok_then/2
	Renamed handle/2 to error_then/2

Added
	retries option for run!/2,run/2, ok_then!/3, and ok_then/3

[0.4.4] - 2025-11-19
Added
	"Interesting Examples" headers

[0.4.3] - 2025-11-19
Added
	More examples in "Interesting Examples" guide

[0.4.2] - 2025-11-19
Changed
	Claude SKILL.md file in docs

Changed
	Docs

[0.4.1] - 2025-11-15
Changed
	"Interesting Examples" guide

[0.4.0] - 2025-11-13
Changed
	Add support for Ecto.Changeset in user_message

[0.3.6] - 2025-11-12
Changed
	Docs

[0.3.5] - 2025-11-12
Changed
	Docs

[0.3.4] - 2025-11-11
Changed
	Docs

[0.3.3] - 2025-11-10
Changed
	Docs

[0.3.2] - 2025-11-10
Changed
	Docs

[0.3.1] - 2025-11-04
Changed
	Docs

[0.3.0] - 2025-11-04
Changed
	map_unless renamed to map_if
	Docs

[0.2.6] - 2025-11-04
Changed
	Docs

[0.2.5] - 2025-11-04
Changed
	Docs

[0.2.4] - 2025-11-03
Changed
	Docs

[0.2.3] - 2025-10-31
Changed
	Docs

[0.2.2] - 2025-10-31
Changed
	Docs

[0.2.1] - 2025-10-31
Changed
	Docs

[0.2.0] - 2025-10-31
Changed
	Rename module to Triage

Added
	Documentation improvements

[0.1.0] - 2025-10-31
Added
	Previous work under a different name, pushed as triage to reserve package



  

    Contexts

Context Wrapping
The wrap_context/3 function adds context (a label and/or metadata) to any errors which come out of a piece of code. This is especially useful for understanding where a failure came from:
defmodule Users do
  alias MyApp.{User, Repo}

  def create_user(params) do
    with :ok <- check_email_availability(params)
         {:ok, changeset} <- User.changeset(%User{}, params)
      Repo.insert(changeset)
    end
    |> Triage.wrap_context("create user", %{email: params[:email]})
  end
end

# When an error occurs, you get a `reason` which is a `Triage.WrappedError` exception struct.
{:error, reason} = Users.create_user(%{name: "Alice", email: "alice@example.com"})

# `Exception.message/1` is a standard Elixir function for getting message strings from exceptions
Exception.message(reason)
# => {:error, #Ecto.Changeset<...>}
#        [CONTEXT] lib/my_app/users.ex:10: create user
You also might find yourself converting an error into another error in order to make it clear what was happening:
(source)
case AdvertisementService.Stub.advertise(channel, request) do
    {:ok, intents} ->
      {:ok, intents}

    {:error, _} ->
      {:error, :failed_to_fetch_intents}
end
This could be a good time to use a context so that the original error is returned along with context label/metadata:
AdvertisementService.Stub.advertise(channel, request)
|> Triage.wrap_context("Fetching advertisement intents", channel_host: channel.host)
Make sure to see the Output section for how wrapped errors can be useful without you needing to work with them directly.


  

    Output

Logging
The log/2 function logs results and passes them through unchanged, making it perfect for debugging pipelines.
Logging Errors Only (the default)
defmodule API.UserController do
  def create(conn, params) do
    Users.create_user(params)
    # Only logs if there's an error
    |> Triage.log()
    # You can also pass :errors (the default)
    # |> Triage.log(:errors)
    |> case do
      {:ok, user} ->
        conn
        |> render("user.json", user: user)
      # Ideally we should return a useful error... see below!
      {:error, _} ->
        conn
        |> put_status(400)
        |> json(%{error: "Unable to create user"})
    end
  end
end
When Users.create_user returns an error, a log is written at the error log level:
# [error] [RESULT] lib/api/user_controller.ex:4: {:error, #Ecto.Changeset<...>}
Logging All Results (:all mode)
In the case above, instead of calling |> log(:errors)  we could call |> log(:all). In that case we could get the error log above, or we could get a success result written to the log at the info level:
# [info] [RESULT] lib/api/user_controller.ex:4: {:ok, %MyApp.Users.User{...}}
Contexts
When errors occur deep in your application's call stack, it can be challenging to understand where your errors are coming from. The wrap_context/3 function allows you to add contextual information at multiple levels, building up a trail of breadcrumbs as errors bubble up.
Here's an example showing how contexts accumulate across different modules:
defmodule MyApp.OrderProcessor do
  def process_payment(order) do
    with {:ok, payment_method} <- fetch_payment_method(order),
         {:ok, charge} <- charge_payment(payment_method, order.amount) do
      {:ok, charge}
    end
    |> Triage.wrap_context("process payment", %{order_id: order.id, order_amount: order.amount})
  end
  # ...
end

defmodule MyApp.OrderService do
  def complete_order(order_id) do
    fetch_order(order_id)
    |> MyApp.OrderProcessor.process_payment(order)
    |> Triage.wrap_context("complete order")
  end
  # ...
end
When an error occurs in the payment processing, logging it will show the full context chain:
def show(conn, %{"order_id" => order_id}) do
  order_id = String.to_integer(order_id)

  MyApp.complete_order(order_id)
  |> Triage.log()
  # ...

# Log output:
# [error] [RESULT] lib/my_app/order_service.ex:15: {:error, :payment_declined}
#     [CONTEXT] lib/my_app/order_service.ex:15: complete order
#     [CONTEXT] lib/my_app/order_processor.ex:8: process payment | %{order_id: 12345, amount: 99.99}
This makes it easy to trace exactly what your application was doing when the error occurred, including both descriptive labels and relevant data.
User-friendly output
It's possible that you might have some code, be it in a LiveView, controller, background worker, etc... Often code at this "top level" might have called a series of functions which call a further series of functions, all of which can potentially return ok/error results.  When getting back {:error, _} tuples specifically, often the value inside of the tuple could be one of many things (e.g. a string/atom/struct/exception, etc...) Often the simplest thing to do is to return something like There was an error: #{inspect(reason)}, but that value often won't make sense to the user.  So we should find a way to make it human-readable, whenever possible.
defmodule MyAppWeb.UserController do
  def checkout(conn, params) do
    MyApp.Users.create_user(params)
    |> case do
      {:ok, result} ->
        conn
        |> render("checkout.json", result: result)
      # Ideally we should return a useful error... see below!
      {:error, _} = error ->
        conn
        |> put_status(400)
        |> json(%{error: Triage.user_message(error)})
    end
    # ...
In this case, you could imagine that MyApp.Users.create_user(params) could return one of the following errors:
# A string
{:error, "Could not contact tracking server"}
# An atom
{:error, :user_not_found}
# A struct containing errors
{:error, %Ecto.Changeset{...}}
# An exception value
{:error, %Jason.DecodeError{...}
Triage.user_message always turns the error into a string and does it's best to extract the appropriate data for a human-readable string.
Additionally, if you use Triage.wrap_context, additional information from the WrappedError will be available to help describe the error.


  

    Enumerating Errors

map_if
source for original example
  defp validate_each_metric(metrics, query) do
    Enum.reduce_while(metrics, [], fn metric, acc ->
      case validate_metric(metric, query) do
        {:ok, metric} -> {:cont, acc ++ [metric]}
        {:error, reason} -> {:halt, {:error, reason}}
      end
    end)
  end
Using Enum.reduce_while is a common pattern when operating over a set of results.  In this case validate_each_metric returns a list if everything is successful while returning the first error if nothing is successful.
We can use the Triage.map_if function to simplify this logic:
  defp validate_each_metric(metrics, query) do
    # Returns {:ok, [...]} where the original returned just [...]
    Triage.map_if(metrics, & validate_metric(&1, query))
  end
In addition to simplifying the code, it's more obvious at first glance what is happening because things have been stripped down to the most important details (map_if is being used on metrics, using validate_metric).
Also, by using triage there is an assurance that we will get a results.
find_value
source for original example
Enum.reduce_while(methods, {:error, :unverified}, fn method, acc ->
  case apply(__MODULE__, method, [sso_domain, domain_identifier, opts]) do
    true -> {:halt, {:ok, method}}
    false -> {:cont, acc}
  end
end)
This is a case where we want to stop on success, not an error.  So we can use Triage.find_value to return the first successful result:
Triage.find_value(methods, fn method ->
  if apply(__MODULE__, method, [sso_domain, domain_identifier, opts]) do
    {:ok, method}
  else
    {:error, :unverified}
  end
end)
Aside from reducing the code down to it's essentials, this also has the advantage of putting the error at the end in the context of the if / else statement.
all
source for original example
  defp validate_list(list, parser_function) do
    Enum.reduce_while(list, :ok, fn value, :ok ->
      case parser_function.(value) do
        :ok -> {:cont, :ok}
        {:error, _} = error -> {:halt, error}
      end
    end)
  end
Here we're looking to see if a function (parser_function) always returns a success or not.  For this we can use Triage.all:
  defp validate_list(list, parser_function) do
    Triage.all(list, parser_function)
  end
Because this was simplified down to a single function call, it would even be possible to get rid of the validate_list function.  It's used in two places:
validate_list(goal_filter_clauses, &validate_goal_filter(&1, configured_goal_names))
with :ok <- validate_list(query.metrics, &validate_metric(&1, query)) do
Which can become:
Triage.all(goal_filter_clauses, &validate_goal_filter(&1, configured_goal_names))
with :ok <- Triage.all(query.metrics, &validate_metric(&1, query)) do
So really in this example, validate_list is just a general function doing the same thing as Triage.all would do


  

    Control Flow

Sometimes you have an result and you want to transform it:
	Performing an actions if the previous was successful
	Returning an error when getting a specific success response (e.g. a HTTP 500)
	Transforming one error into another
	Ignoring an error result (i.e. turning it into a success)

In these cases you can use Triage.ok_then, Triage.ok_then! to error_then :ok results and Triage.error_then to error_then :error results.
The ok_then functions work under the assumption that you're going to return another success, so whatever result you return will be wrapped in an {:ok, _} tuple when returned.  Similarly, error_then will wrap the result in an {:error, _} tuple.  But if you return :error results from ok_then or :ok results from error_then then no wrapping occurs. This is so that you can have a flow where changes from the norm (:ok -> :error or :error -> :ok) stand out from the rest of the flow.
run and ok_then functions
The ok_then functions provide a way to chain operations that return results. They allow you to build pipelines of transformations where errors automatically short-circuit the chain.
run!/1 - Execute a function + error_then exceptions
Takes a zero-arity function and executes it.  The function can return :ok, {:ok, term()}, :error, or {:error, term()}, but any other value is treated as {:ok, <value>}.
# order_id is defined / passed in
Triage.run(fn -> fetch_order_from_api(order_id) end)
run/1 - Execute a function + error_then exceptions
Works like run!/1, but if an exception is raised then a {:ok, WrappedError.t()} is returned which wraps the exception.
ok_then!/2 - Chaining operations + allowing exceptions to raise
Takes a result and a function, executing the function only if the result is successful. Unlike ok_then/2, this does not catch exceptions:
# Example: User registration pipeline
fetch_order_from_api(order_id)
|> Triage.ok_then!(&validate_order/1)
|> Triage.ok_then!(fn order -> change_for_order(order, user.payment_info) end)
# => {:ok, user} if all thens succeed
# If any ok_then returns an error, further thens are ignored and the error is passed through

# When given :ok, passes nil to the function
# Previous ok_then returns `:ok`
|> Triage.ok_then!(fn nil -> send_notification() end)
# => {:ok, notification_result}
ok_then/2 - Chaining operations + handling exceptions
Behaves like ok_then!/2 but catches exceptions and wraps them in a WrappedError:
# Example: Processing an API response
{:ok, response}
|> Triage.ok_then(fn response -> Jason.decode!(response.body) end)  # Might raise
|> Triage.ok_then(&validate_schema/1)
|> Triage.ok_then(&transform_data/1)
# => {:ok, transformed_data}

# Catches exceptions during parsing
{:ok, config_string}
|> Triage.ok_then(&String.to_integer/1)  # Raises if not a valid integer
|> Triage.ok_then(&update_config/1)      # Never called if parsing raises
|> Triage.log()
Log output when String.to_integer/1 raises:
[error] [RESULT] lib/my_app/config.ex:42: ** (ArgumentError) errors were found at the given arguments:

 * 1st argument: not a textual representation of an integer

    [CONTEXT] :erlang.binary_to_integer/1
error_then
The Triage.error_then/2 function takes in a result and uses a callback function to determine how error results should be handled, passing ok results through unchanged.
# Imagine this function returns either:
#  * {:ok, Confirmation.t()}
#  * {:error, :invoice_invalid}
#  * {:error, :credit_card_expired}
#  * {:error, :billing_service_down}
bill_customer(customer, invoice)
|> Triage.error_then(fn
  :invoice_invalid ->
    {:invoice_invalid, invoice_errors(invoice)}

  :billing_service_down ->
    # Queue for billing later

    {:ok, :bill_later}

  other ->
    other
end)
# Returns:
#  * {:ok, Confirmation.t()}
#  * {:ok, :bill_later}
#  * {:error, {:invoice_invalid, ...}}
#  * {:error, :credit_card_expired}
An exception with a function which returns an error result with an exception reason:
Jason.decode(string)
# Jason returns a `Jason.DecodeError` exception struct
# Here we call Elixir's `Exception.message/1` to turn it into a string
|> Triage.error_then(fn error -> Exception.message(error) end)
You can use Triage.error_then/2 to transform the error based on pattern matching. Here's an example combining ok_then and error_then:
HTTPoison.get(url)
|> Triage.ok_then(fn
  %HTTPoison.Response{status_code: 200, body: body} ->
    body

  %HTTPoison.Response{status_code: 404, body: body} ->
    {:error, "Server result not found"}
end)
|> Triage.error_then(fn
    %HTTPoison.Error{reason: :nxdomain} ->
      "Server domain not found"

    %HTTPoison.Error{reason: :econnrefused}
      "Server connection refused"

    %HTTPoison.Error{reason: reason}
      "Unexpected error connecting to server: #{inspect(reason)}"
end)
Or you can provide a default value on failure:
Jason.decode(string)
|> Triage.error_then(fn _ -> {:ok, @default_result} end)
[!NOTE]
You might be wondering why you might use ok_then / error_then functions instead of with.  If so, check out the Comparison to with section.

tap_ok and tap_error
The Triage.tap_ok/2 and Triage.tap_error/2 functions allow you to perform side effects (like logging, notifications, or analytics) without changing the result. Unlike ok_then and error_then, these functions always pass the original result through unchanged, making them ideal for observation and monitoring.
tap_ok/2 - Side effects on success
Executes a function only for successful results, passing through the original result unchanged:
# Example: Track analytics and send notifications for user registrations
register_user(params)
|> Triage.tap_ok(fn user ->
  Analytics.track("user_registered", %{user_id: user.id})
  NotificationService.send_welcome_email(user)
end)
|> Triage.ok_then(&add_to_default_groups/1)
# => {:ok, user} - the tap doesn't affect the result

# Example: Caching successful API responses
fetch_product_from_api(product_id)
|> Triage.tap_ok(& Cache.put("product:#{product_id}", &1))
|> Triage.ok_then(&transform_product/1)
# The cache is populated but the result continues through the pipeline
When given :ok, the function receives nil:
delete_user(user_id)
# Returns :ok on success
|> Triage.tap_ok(fn nil ->
  AuditLog.record_deletion(:user, user_id)
end)
# => :ok
tap_error/2 - Side effects on failure
Executes a function only for error results, passing through the original error unchanged:
# Example: Error monitoring and alerting
process_payment(order)
|> Triage.tap_error(fn error ->
  ErrorTracker.report(error, context: %{order_id: order.id})

  if error == :payment_gateway_down do
    PagerDuty.alert("Payment gateway is down")
  end
end)
|> Triage.error_then(&handle_payment_error/1)
# => {:error, reason} - the tap doesn't change the error

# Example: Track metrics for failed operations
update_inventory(product_id, quantity)
|> Triage.tap_error(fn reason ->
  Metrics.increment("inventory.update.failed", tags: %{reason: reason})
  ErrorTracker.report(reason, context: %{product_id: product_id})
end)
# => {:error, reason}
When given :error, the function receives nil:
validate_required_fields(params)
# Returns :error when validation fails with no specific reason
|> Triage.tap_error(fn nil ->
  Metrics.increment("validation.failed.no_reason")
end)
# => :error


  

    Comparison to with

Because this package sometimes deals with sequences steps which return ok/error results, a common question is "why not just use the with clause?". The truth is, sometimes with is great, but most people assume that with is for handling ok/error results when that is just the most common use-case.  with is a more general tool and if you don't understand how it works, it may have unexpected behavior.
Here is an example that demonstrates a few potential gotchas with the with clause:
with {:ok, a} <- function1(...),
     {:ok, %{"b" => b} <- function2(a),
     {:ok, c} <- function3(b) do
  # ...
end
Let's imagine a few things that might happen which could be unexpected.  Imagine that:

function1 sometimes returns {:ok, _} and :ok other times
You might imagine that you'd get a MatchError, but actually the whole with clause will simply return :ok without running function2, function3, or the body.
You might think it would be weird for a function to return inconsistent :ok results, but it could be that function1 returns the result of multiple functions that it's calling.

function2 sometimes returns {:ok, map()} but the map doesn't have a "b" key
Again, the with will return {:ok, <map>} without running function3 or the body.
This could happen particularly when the function is getting a JSON response from a server and your app has no control over that response.

function2 returns :error but function1 and function3 return {:error, reason}
This means that your with is going to return an inconsistent error result.
You could just introduce an else
with {:ok, a} <- function1(...),
     {:ok, %{"b" => b} <- function2(a),
     {:ok, c} <- function3(b) do
  # ...
else
  :ok ->
    # ...

  {:ok, value} ->
    # ...

  :error ->
    {:error, ...}
end
This gets into the Complex else clauses in with anti-pattern where it becomes hard to keep track of which else clause handles which with clause. Also to make sure you've covered all cases you need to dig into function1/function2/function3, which often isn't trivial.
Real Examples
A Simple Example
Let's take an example from the "Complex else clauses in with" anti-pattern documentation. The following two functions were extracted from with clauses to make sure that the responses are standardized across clauses:
defp file_read(path) do
  case File.read(path) do
    {:ok, contents} -> {:ok, contents}
    {:error, _} -> {:error, :badfile}
  end
end

defp base_decode64(contents) do
  case Base.decode64(contents) do
    {:ok, decoded} -> {:ok, decoded}
    :error -> {:error, :badencoding}
  end
end
This is an excellent idea and it's one of the problems triage tries to solve.
We can imagine refactoring these functions to use further with clauses:
defp file_read(path) do
  with {:error, _} <- File.read(path) do
    {:error, :badfile}
  end
end

defp base_decode64(contents) do
  with :error <- Base.decode64(contents) do
    {:error, :badencoding}
  end
end
This has the advantage of making it very explicit which pattern we're doing something with and which pattern we're just passing through unchanged.
On the other hand, some people might not be comfortable:
	...using with in a "non-standard" way (matching on :error results instead of :ok results)
	...using with with just one clause
	...not being explicit with each case (and, to be fair, there is sometimes value in a case raising a MatchError  if a path isn't covered)

defp file_read(path) do
  File.read(path)
  |> Triage.error_then(fn _ -> :badfile end)
end

defp base_decode64(contents) do
  Base.decode64(contents)
  # Note: error_then/2 receives `nil` for bare `:error` atom results
  |> Triage.error_then(fn nil -> :badencoding end)
end
Here we simplify the error handling logic and move it to the end so it reads top-to-bottom. We also get similar behavior to case where we get a MatchError if a pattern isn't accounted for.
A More Complex Example
Finally, let's take an example from oban's source code:
  defp insert_unique(conf, changeset, opts) do
    opts = Keyword.put(opts, :on_conflict, :nothing)

    with {:ok, query, lock_key} <- unique_query(changeset),
         :ok <- acquire_lock(conf, lock_key),
         {:ok, job} <- fetch_job(conf, query, opts),
         {:ok, job} <- resolve_conflict(conf, job, changeset, opts) do
      {:ok, %{job | conflict?: true}}
    else
      {:error, :locked} ->
        with {:ok, job} <- Changeset.apply_action(changeset, :insert) do
          {:ok, %{job | conflict?: true}}
        end

      nil ->
        Repo.insert(conf, changeset, opts)

      error ->
        error
    end
  end
This definitely requires a lot from the reader if they need to figure out which else clause matches up with which with clause. And, in fact, there are two cases which return nil. If we dig down into each function which is called in the with clauses we find:
	unique_query returns {:ok, _, _} or nil
	acquire_lock returns :ok or {:error, :locked}
	fetch_job returns {:ok, job} or nil
	resolve_conflict returns {:ok, job}, {:ok, Ecto.Schema.t()}, {:error, Ecto.Changeset.t()}

Aside from having a complex else clause, two of the functions return nil in some cases. If a function returns :ok / {:ok, _}, it's clearer to make sure it's always returning some sort of :ok / :error result. In this case we could return {:error, :not_found} to indicate the failure. So, how might we put this code another way? We can use the Triage.ok_then!/2 and Triage.error_then/2 functions, which work with :ok and :error results, respectively:
defp insert_unique(conf, changeset, opts) do
  opts = Keyword.put(opts, :on_conflict, :nothing)

  # Need to refactor `unique_query` to return `{:ok, {query, lock_key}}`
  unique_query(changeset)
  |> Triage.ok_then!(fn {query, lock_key} ->
    acquire_lock(conf, lock_key)
    |> Triage.error_then(fn
      :locked ->
        Changeset.apply_action(changeset, :insert)
        |> Triage.ok_then!(& %{&1 | conflict?: true})
    end)
    |> Triage.ok_then!(fn _ -> fetch_job(conf, query, opts) end)
  end)
  |> Triage.ok_then!(& resolve_conflict(conf, &1, changeset, opts))
  # Assuming we refactor `unique_query` and `fetch_job` to return `{:error, :not_found}` instead of `nil`
  |> Triage.ok_then!(fn job -> %{job | conflict?: true} end)
  |> Triage.error_then(fn :not_found -> {:ok, Repo.insert(conf, changeset, opts)} end)
end
At first glance this doesn't seem as clean because it doesn't have all of the happy-path cases followed by all of the error handling. But this version has several advantages:
	We're focused on the values from the {:ok, _} / {:error, _} tuples without dealing with pattern matching on them.
	Tuples are only specified when we're turning a success into an error or vice-versa which makes those special cases stand out.
	Error handling is done at the soonest point that it can be handled (just after the call or after calls that might share the same error).
	Nesting makes it clear where the query variable is needed.
	We can use ok_then (not used above, but used instead of ok_then!) to catch errors, if we don't want a particular step to crash.


Am I trying to convince you to use this?  Kind of... but only if it makes sense for the situation!


  

    Claude SKILL

Claude SKILL
Since triage is a separate library, LLMs might not have a lot of experience working with it. The following is a Claude SKILL.md file (under development still) that people can use to improve Claude's ability to work with triage. It may also help train other LLMs to use triage.
See the latest SKILL.md file in the repo if you'd like to add it to your own environment.


  

    Interesting Examples

A single case statement
Source from the mindwedel project.
This code is part of a larger function which makes an LLM request:
case ChatCompletions.create(openai_client, chat_completion) do
  {:ok, response} when is_map(response) ->
    {:ok, response}

  {:error, %OpenAIError{status_code: status, message: message} = error} ->
    Logger.error("""
    OpenAI API error:
    Status: #{status}
    Message: #{message || "No message"}
    Full error: #{inspect(error)}
    Provider: #{ai_config[:provider]}
    Model: #{ai_config[:model]}
    """)

    {:error, :llm_request_failed}

  {:error, error} ->
    Logger.error("Unexpected OpenAI API error: #{inspect(error)}")

    {:error, :llm_request_failed}
end
	Since there is just one :ok pattern, we can simplify it with a single ok_then! call.
	While the wrap_context won't output the same log exactly, it will provide metadata for either :error case
	Since we always return :llm_request_failed for errors, this can be a single error_then

Since the purpose of returning :llm_request_failed is probably to help locate the error when there's a failure, we could even skip the error_then entirely and error_then the WrappedError higher up which will have information about where the error came from.
ChatCompletions.create(openai_client, chat_completion)
|> Triage.ok_then!(fn response when is_map(response) -> response end)
|> Triage.wrap_context(
  status: status,
  message: message,
  provider: ai_config[:provider],
  model: ai_config[:model]
)
|> Triage.log()
|> Triage.error_then(fn _ -> :llm_request_failed end)
Two functions, two cases
Source from the anoma project.
  def store_file(filename, file_path, content_type, s3_client \\ storage_provider()) do
    case File.read(file_path) do
      {:ok, file} ->
        case Vault.encrypt(file) do
          {:ok, encrypted_file} ->
            store_encrypted_file(filename, encrypted_file, content_type, s3_client)

          {:error, error_message} ->
            {:error, "Issue while encrypting file: #{inspect(error_message)}"}
        end

      {:error, reason} ->
        Logger.error("Failed to read file #{file_path}: #{inspect(reason)}")
        {:error, "Failed to read file: #{inspect(reason)}"}
    end
  end

  defp store_encrypted_file(filename, encrypted_file, content_type, s3_client) do
    encrypted_file_path = upload_path(filename)

    case s3_client.put_object(
           bucket_name(),
           encrypted_file_path,
           encrypted_file,
           %{
             content_type: content_type
           }
         ) do
      {:ok, _headers} ->
        {:ok, encrypted_file_path}

      {:error, {error_type, http_status_code, response}} ->
        Logger.error(
          "Error storing file in bucket: #{filename} Type: #{content_type}. Error type: #{error_type} Response code: #{http_status_code} Response Body: #{response.body}"
        )

        {:error, "Issue while storing file."}
    end
  end
This allows us to reduce 39 lines down to 20. Also, since store_encrypted_file is only used inside of store_file, so collapsing it's logic allows us to combine it all into store_file.
If we use Triage.log just after File.read we log just those errors like the original version, but putting it at the end lets us capture everything. The two wrap_context calls also allow us to isolate errors when they happen, giving relevant metadata for logging depending on the location.
  def store_file(filename, file_path, content_type, s3_client \\ storage_provider()) do
    File.read(file_path)
    |> Triage.ok_then!(&Vault.encrypt/1)
    |> Triage.ok_then!(fn encrypted_file ->
      encrypted_file_path = upload_path(filename)

      s3_client.put_object(
        bucket_name(),
        encrypted_file_path,
        encrypted_file,
        %{content_type: content_type}
      )
      |> Triage.wrap_context("Putting to S3", filename: filename)
      |> Triage.ok_then!(fn _ -> encrypted_file_path end)
    end)
    |> Triage.wrap_context(
      "storing encrypted file",
      filename: filename, file_path: file_path, content_type: content_type
    )
    |> Triage.log()
  end
Large and complex example
Source from the anoma project.
@doc lines removed for brevity
  @spec compose([binary()]) ::
          {:ok, binary()}
          | {:error, :invalid_input, term()}
          | {:error, :noun_not_a_valid_transaction}
          | {:error, :not_enough_transactions}
  def compose(transactions) do
    # fetch the jammed intents from the request
    with {:ok, nouns} <- cue_transactions(transactions),
         {:ok, transactions} <- nouns_to_transactions(nouns),
         {:ok, composed} <- compose_transactions(transactions),
         noun <- Nounable.to_noun(composed),
         jammed <- Jam.jam(noun) do
      {:ok, jammed}
    else
      {:error, :cue_failed, err} ->
        {:error, :invalid_input, err}

      {:error, :noun_not_a_valid_transaction} ->
        {:error, :noun_not_a_valid_transaction}

      {:error, :not_enough_transactions} ->
        {:error, :not_enough_transactions}
    end
  end

  @spec verify(binary()) ::
          {:ok, boolean()}
          | {:error, :noun_not_a_valid_transaction | :verify_failed}
          | {:error, :cue_failed, term()}
  def verify(transaction) do
    with {:ok, noun} <- cue_transaction(transaction),
         {:ok, transaction} <- noun_to_transaction(noun),
         valid? when is_boolean(valid?) <- Transaction.verify(transaction) do
      {:ok, valid?}
    else
      {:error, :cue_failed, err} ->
        {:error, :cue_failed, err}

      {:error, :noun_not_a_valid_transaction} ->
        {:error, :noun_not_a_valid_transaction}
    end
  end

  @spec cue_transactions([binary()]) ::
        {:ok, [Noun.t()]} | {:error, :cue_failed, term()}
  defp cue_transactions(transactions) do
    Enum.reduce_while(transactions, [], fn tx, acc ->
      case cue_transaction(tx) do
        {:ok, noun} ->
          {:cont, [noun | acc]}

        {:error, :cue_failed, err} ->
          {:halt, {:error, :cue_failed, err}}
      end
    end)
    |> case do
      {:error, :cue_failed, err} ->
        {:error, :cue_failed, err}

      txs ->
        {:ok, txs}
    end
  end

  @spec cue_transaction(binary()) ::
        {:ok, Noun.t()} | {:error, :cue_failed, term()}
  defp cue_transaction(transaction) do
    case Jam.cue(transaction) do
      {:ok, noun} ->
        {:ok, noun}

      {:error, %{message: err}} ->
        {:error, :cue_failed, err}
    end
  end

  @spec nouns_to_transactions([Noun.t()]) ::
          {:ok, [Transaction.t()]} | {:error, :noun_not_a_valid_transaction}
  defp nouns_to_transactions(nouns) do
    Enum.reduce_while(nouns, [], fn tx, acc ->
      case noun_to_transaction(tx) do
        {:ok, transaction} ->
          {:cont, [transaction | acc]}

        {:error, :noun_not_a_valid_transaction} ->
          {:halt, {:error, :noun_not_a_valid_transaction}}
      end
    end)
    |> case do
      {:error, :noun_not_a_valid_transaction} ->
        {:error, :noun_not_a_valid_transaction}

      txs ->
        {:ok, txs}
    end
  end

  @spec noun_to_transaction(Noun.t()) ::
          {:ok, Transaction.t()} | {:error, :noun_not_a_valid_transaction}
  defp noun_to_transaction(noun) do
    case Transaction.from_noun(noun) do
      {:ok, transaction} ->
        {:ok, transaction}

      :error ->
        {:error, :noun_not_a_valid_transaction}
    end
  end
Aside from generally removing the boilerplate of handling {:ok, _} and {:error, _} wrappers, this refactor:
	... removes 52 lines of the original 108 lines code (48%)
	... removes two functions
	... makes it clear at a higher level when we're mapping over operations

If we moved to using Triage.wrap_context and changed the FallbackController to error_then the resulting WrappedErrors, we could also potentially remove some of the error handling here while also adding some useful context to errors which are returned.
  @spec compose([binary()]) ::
          {:ok, binary()}
          | {:error, :invalid_input, term()}
          | {:error, :noun_not_a_valid_transaction}
          | {:error, :not_enough_transactions}
  def compose(transactions) do
    # fetch the jammed intents from the request
    transactions
    # Ordered doesn't matter for these two lines because the transactions
    # are going to be composed so while this will produce the reverse
    # of the original, it should be fine
    |> Triage.map_if(&cue_transaction/1)
    |> Triage.map_if(&noun_to_transaction/1)
    |> Triage.ok_then!(&compose_transactions/1)
    |> Triage.ok_then!(fn composed ->
      composed
      |> Nounable.to_noun()
      |> Jam.jam()
    end)
    |> Triage.error_then(fn
      {:cue_failed, err} ->
        {:invalid_input, err}

      :noun_not_a_valid_transaction ->
        :noun_not_a_valid_transaction

      :not_enough_transactions ->
        :not_enough_transactions
    end)
  end

  @spec verify(binary()) ::
          {:ok, boolean()}
          | {:error, :noun_not_a_valid_transaction | :verify_failed}
          | {:error, :cue_failed, term()}
  def verify(transaction) do
    cue_transaction(transaction)
    |> Triage.ok_then!(&noun_to_transaction/1)
    |> Triage.ok_then!(fn transaction ->
      # will raise a `MatchError` when not a boolean
      # should be basically the same result as the `WithClauseError`
      # which would have been raised before
      case Transaction.verify(transaction) do
        valid? when is_boolean(valid?) ->
          valid?
      end
    end)
  end

  @spec cue_transaction(binary()) ::
        {:ok, Noun.t()} | {:error, :cue_failed, term()}
  defp cue_transaction(transaction) do
    Jam.cue(transaction)
    |> Triage.error_then(fn %{message: err} -> {:cue_failed, err} end)
  end

  @spec noun_to_transaction(Noun.t()) ::
          {:ok, Transaction.t()} | {:error, :noun_not_a_valid_transaction}
  defp noun_to_transaction(noun) do
    Transaction.from_noun(noun)
    |> Triage.error_then(fn :error -> :noun_not_a_valid_transaction end)
  end


  

    Philosophy

Standard results (:ok, :error, {:ok, term()}, {:error, term()})
ok/error results tend to get returned and used throughout a codebase, so having a limited set of possible combinations means there are fewer patterns to expect.
Focus on the most useful tools
This library came out of a general desire of seeing the same need for patterns over and over. But in choosing which functions to create, the use-case must exist multiple times in a real-life project. You will find that many of the examples in the docs come from real-life examples.
The Principle of Least Surprise
As much as possible tools in the triage library should be named according to the principle of least surprise. For example, Triage.map_if is named because it does a map based on iflogic, Triage.find_value is meant to work like Enum.find_value, and  Triage.ok_then/Triage.error_then are meant to work similarly to Elixir's Kernel.then.
Because naming is so important, feedback and ideas on the API are very welcome!
Prefer 2-element result tuples
Using 2-element result tuples ({:ok, term()} / {:error, term()}) is preferable because it makes for more standard patterns to match against as errors are returned and combined up the stack.  But not all tools return 2-element result tuples.  So the triage library supports longer result tuples when doing things like logging, but whenever it returns a result it is always 2-element results.
Variety of small tools that work together (UNIX philosophy)
Triage isn't meant to be the ultimate tool to be used for all code which might return errors. Because we're working with standard results that already exist, you should be able to slip in triage tools as needed along side case, with, and other standard Elixir patterns.
Avoid macros
Macros might be useful for some specific tools, but the base assumption is that there should be plain functions available to use if desired.
Aside from potential complexity in debugging, you should be able to drop Triage.* calls into your code without needing to think if you've added a require.


  

    Open Questions

There are a lot of design considerations that go into the triage library. There are still some open questions to be considered. Feel free to open an issue or a discussion on the GitHub repo.
Should other atoms / tuples be supported?
Sometimes "results" can be more than just :ok / :error.  Common examples: :ignore or :commit (both alone and as the label in a tuple).
Should triage support these in some way (limited or otherwise)? Should it be something an application can configure?
An argument could be make that one could return things like {:ok, :ignore}, {:error, :ignore}, {:ok, {:commit, _}}
It also shouldn't be the goal that people use triage for all types of result objects, so being limited to :ok / :error results may be just fine.
Bang syntax
I'm curious what people think about functions like ok_then vs ok_then!.  The ok_then! could be considered a "default" because it doesn't "do" anything. Rather it's the ok_then function which actively rescues from exceptions. But it does feel right, in a way, to have a simple convention that ok_then! is the one where you can expect exception to come from 🤷‍♂️
Should ok_then return WrappedError?
Currently when an exception is raised in the function given to Triage.ok_then, it returns a WrappedError which "contains" the exception. This WrappedError automatically includes context information about the function (the line/number if anonymous and the name of the function if captured). Should the raw exception be returned instead? It wouldn't have STACKTRACE information because that's not on exceptions.
Should ok_then!/1 exist?
Calling ok_then! functions doesn't catch exceptions.  So is there a difference between calling ok_then!/1 with some code or just running that code directly?
	ok_then/1 captures errors
	ok_then!/2 / ok_then/2 only execute the function on :ok results and unwrap the result

The ok_then!/1 function at least has the behavior that you can return a non-result and it will automatically wrap it in {:ok, _}, but is that worthwhile enough?


  

    Logging JSON

By default, logs are formatted as human-readable plain text. If you would like to output JSON logs, you can use a library like logger_json. The Triage.log function sets the errors_result_details metadata key as well as setting metadata given by wrap_context calls.  You can set these keys as output in your logger configuration.  The errors_result_details key gives nested a key/value structure, so it won't be outputted with default logs and makes sense when outputting structured logs like with json.
Here is an example of configuring metadata:
# config/config.exs
config :logger, :console,
 format: "[$level] $message $metadata\n",
 metadata: [:user_id]
To configure logger_json, you might use something like this:
config :logger, :default_handler,
  formatter:
    LoggerJSON.Formatters.Basic.new(metadata: [:user_id, :errors_result_details])
It is recommended to put this in your config/runtime.exs under if config_env() == :prod do so that you can get more readable logs in dev and test environments.
If you were to use Triage.wrap_context("updating user", user_id: 123):
With standard logging you'd get user_id=123 just like if you gave the metadata to Logger.error yourself.
Here is an example of what you might get with logger_json (spacing introduced for readability):
{
  "message": "[RESULT] {:error, :not_found}\n    [CONTEXT] updating user %{user_id: 123}",
  "time": "2025-10-24T13:20:06.885Z",
  "metadata": {
    "errors_result_details": {
      "reason": "not_found",
      "type": "error"
    },
    "user_id": 123
  },
  "severity": "error"
}


  

    Configuration

app
Because of how tail-call optimisation affects stack-traces in Elixir, logging a result may not give the right line.  An example is when you give an anonymous function to a library:
defmodule MyAppWeb.UserController do
  def create(conn, params) do
    SomeLibrary.execute(fn ->
      MyApp.Users.create_user(params)
      |> Triage.log()
    end)
  end
end


# The line in the log might look something like:
# [error] [RESULT] lib/some_library.ex:4: {:error, #Ecto.Changeset<...>}

# Where you would like it to show:
# [error] [RESULT] lib/my_app_web/user_controller.ex:5: {:error, #Ecto.Changeset<...>}
In order to help log the correct entries from the stacktrace, you can optionally configure your app's name to help it be found:
# config/config.exs
config :triage, :app, :your_app_name


  

    find_working_url

Source for original code
The following case uses reduce_while to check two domains via DNS:
  # Check A records of the the domains [domain, "www.#{domain}"]
  # at this point, domain can contain path
  @spec find_working_url(String.t()) :: {:ok, String.t()} | {:error, :domain_not_found}
  defp find_working_url(domain) do
    [domain_without_path | rest] = split_domain(domain)

    [
      domain_without_path,
      "www.#{domain_without_path}"
    ]
    |> Enum.reduce_while({:error, :domain_not_found}, fn d, _acc ->
      case dns_lookup(d) do
        :ok -> {:halt, {:ok, "https://" <> unsplit_domain(d, rest)}}
        {:error, :no_a_record} -> {:cont, {:error, :domain_not_found}}
      end
    end)
  end
Using find_value, ok_then, and error_then we can reduce the reduce_while boilerplate:
  # Check A records of the the domains [domain, "www.#{domain}"]
  # at this point, domain can contain path
  @spec find_working_url(String.t()) :: {:ok, String.t()} | {:error, :domain_not_found}
  defp find_working_url(domain) do
    [domain_without_path | rest] = split_domain(domain)

    [
      domain_without_path,
      "www.#{domain_without_path}"
    ]
    |> Triage.find_value(&dns_lookup/1)
    |> Triage.ok_then(& "https://" <> unsplit_domain(&1, rest))
    |> Triage.error_then(fn :no_a_record -> :domain_not_found end)
  end

        wrap_context(result, context)

      


        Wraps result/0 with additional context information, leaving :ok results unchanged.



    


    
      
        wrap_context(result, context, metadata)

      


        Wrap errors from a result with both a context string and metadata. See wrap_context/2



    





  
    Functions > Control Flow
  


    
      
        error_then(result, func)

      


        For dealing with :error cases, passing :ok results through unchanged.



    


    
      
        ok_then(result, func, opts \\ [])

      


        Executes a function with a previous result value, with exception handling.



    


    
      
        ok_then!(result, func, opts \\ [])

      


        Executes a function with a previous result value, without exception handling.



    


    
      
        run(func, opts \\ [])

      


        Executes a function that returns a result tuple, with exception handling.



    


    
      
        run!(func, opts \\ [])

      


        Executes a function that returns a result tuple, without exception handling.



    


    
      
        tap_error(result, func)

      


        Executes a side-effect function when the result is an error, passing the result through unchanged.



    


    
      
        tap_ok(result, func)

      


        Executes a side-effect function when the result is :ok, passing the result through unchanged.

        all(error, func)

      


        Validates that a callback function gives an :ok / {:ok, _} result for all elements
in the given enumerable.



    


    
      
        find_value(error, func)

      


        Finds the first successful result from applying a function to enumerable elements.



    


    
      
        map_if(error, func)

      


        Maps a function over an enumerable, collecting successful values and short-circuiting on the first error.

        error?(result)

      


        Checks if a result is an error (:error or {:error, any()}).



    


    
      
        log(result, mode \\ :errors)

      


        Logs a result tuple and returns it unchanged.



    


    
      
        ok?(result)

      


        Checks if a result is a success (:ok or {:ok, any()}).



    


    
      
        user_message(arg)

      


        Generates a user-friendly error message from various error types.

          @type result() :: :ok | {:ok, any()} | :error | {:error, any()}

          @type value_result() :: {:ok, any()} | :error | {:error, any()}

          @spec wrap_context(result(), String.t() | keyword() | map()) ::
  :ok | {:ok, any()} | {:error, Triage.WrappedError.t()}


      


Wraps result/0 with additional context information, leaving :ok results unchanged.
Takes a result tuple and wraps error cases (:error or {:error, reason}) with
a context string, metadata, and stacktrace info contained in Triage.WrappedError{}.
If the second argument is a string, the context is set. If the second argument is a
keyword list or a map the metadata is set.  The arity 3 version allows setting both.
Because the context is designed to show what was happening in logs or in user error
messages, the context string should describe what is being attempted in business logic
terms. e.g.:
	GOOD: "connecting to user service" (describe the action)
	BAD: "failed to connect to user service" ("failed to connect" describes a failure)
	BAD: "HTTP request to user_authentication server" (technical terms)

The log/2 and user_message/1 functions support WrappedError results. See their
docs for more details
Examples
iex> Triage.wrap_context(:ok, "fetching user")
:ok

iex> Triage.wrap_context({:ok, 42}, "fetching user")
{:ok, 42}

iex> Triage.wrap_context(:error, "fetching user")
{:error, %Triage.WrappedError{}}

iex> Triage.wrap_context({:error, :not_found}, "fetching user", %{user_id: 123})
{:error, %Triage.WrappedError{}}

iex> Triage.wrap_context({:error, :not_found}, %{user_id: 123})
{:error, %Triage.WrappedError{}}

  



  
    
      
    
    
      wrap_context(result, context, metadata)



        
          
        

    

  


  

      

          @spec wrap_context(result(), String.t() | nil, keyword() | map()) ::
  :ok | {:ok, any()} | {:error, Triage.WrappedError.t()}


      


Wrap errors from a result with both a context string and metadata. See wrap_context/2

      error_then(result, func)



        
          
        

    

  


  

      

          @spec error_then(result(), (any() -> any())) :: result()


      


For dealing with :error cases, passing :ok results through unchanged.
When given result is {:error, reason}, the reason is passed into the callback
function. The callback function can then return a new reason which will be
returned from error_then wrapped in an {:error, _} tuple.
If :error is the given result, nil will be given to the callback function.
The callback function can also return :ok or {:ok, any()} to have the error
be ignored and the :ok result will be returned instead.
Examples
iex> ping_account_server() |> Triage.error_then(fn _ -> :account_server_failure end)
{:error, :account_server_failure}

iex> Triage.error_then({:error, :unknown}, fn :unknown -> {:ok, @default_value} end)
{:ok, ...}

iex> Triage.error_then(:ok, fn _ -> :not_used end)
:ok

iex> Triage.error_then({:ok, ...}, fn _ -> :not_used end)
{:ok, 42}

iex> Triage.error_then(:error, fn nil -> :handled end)
{:error, :handled}

  



    

  
    
      
    
    
      ok_then(result, func, opts \\ [])



        
          
        

    

  


  

      

          @spec ok_then(result(), (any() -> any()), [{:retries, non_neg_integer()}]) :: result()


      


Executes a function with a previous result value, with exception handling.
Takes a result and, if :ok, passes the value from the tuple to the provided function.
If the previous result was an error, short-circuits and returns the error.
If the function raises an exception, it catches it and returns
{:error, %Triage.WrappedError{}} with details about the exception.
Parameters
	result - The previous result (:ok, {:ok, value}, :error, or {:error, reason})
	func - A function that takes the unwrapped value and returns a result

Examples
iex> ok_then({:ok, 5}, fn x -> {:ok, x * 2} end)
{:ok, 10}

iex> ok_then({:ok, 5}, fn _x -> raise "boom" end)
{:error, %Triage.WrappedError{}}

  



    

  
    
      
    
    
      ok_then!(result, func, opts \\ [])



        
          
        

    

  


  

      

          @spec ok_then!(result(), (any() -> any()), [{:retries, non_neg_integer()}]) ::
  result()


      


Executes a function with a previous result value, without exception handling.
Takes a result and, if it's an :ok, passes the value from the tuple to the provided function.
If the result was an error, short-circuits and returns the error without calling the function.
This is the "unsafe" version that doesn't catch exceptions. Use ok_then/2 for
exception handling.
Parameters
	result - The previous result (:ok, {:ok, value}, :error, or {:error, reason})
	func - A function that takes the unwrapped value and returns a result

Examples
iex> ok_then!({:ok, 5}, fn x -> {:ok, x * 2} end)
{:ok, 10}

iex> ok_then!({:error, :not_found}, fn x -> {:ok, x * 2} end)
{:error, :not_found}

  



    

  
    
      
    
    
      run(func, opts \\ [])



        
          
        

    

  


  

      

          @spec run((-> any()), [{:retries, non_neg_integer()}]) :: result()


      


Executes a function that returns a result tuple, with exception handling.
Calls the provided zero-arity function and ensures it returns a valid result.
If the function raises an exception, it catches it and returns
{:error, %Triage.WrappedError{}} with details about the exception.
Parameters
	func - A zero-arity function that returns a result

Examples
iex> run(fn -> {:ok, 42} end)
{:ok, 42}

iex> run(fn -> raise "boom" end)
{:error, %Triage.WrappedError{}}

  



    

  
    
      
    
    
      run!(func, opts \\ [])



        
          
        

    

  


  

      

          @spec run!((-> any()), [{:retries, non_neg_integer()}]) :: result()


      


Executes a function that returns a result tuple, without exception handling.
Calls the provided zero-arity function and checks that it returns a result
(:ok, {:ok, value}, :error, or {:error, reason}). If the function returns
any other value, it wraps it in {:ok, value}.
This is the "unsafe" version that doesn't catch exceptions. Use run/1 for
exception handling.
Parameters
	func - A zero-arity function that returns a result

Examples
iex> run!(fn -> 42 end)
{:ok, 42}

iex> run!(fn -> {:ok, 42} end)
{:ok, 42}

iex> run!(fn -> {:error, :not_found} end)
{:error, :not_found}

iex> run!(fn -> :error end)
:error

iex> run!(&function_to_try, retries: 2)
:error

  



  
    
      
    
    
      tap_error(result, func)



        
          
        

    

  


  

      

          @spec tap_error(result(), (any() -> any())) :: result()


      


Executes a side-effect function when the result is an error, passing the result through unchanged.
Takes a result and, if it's :error or {:error, reason}, calls the provided function
with the unwrapped reason (or nil for :error). The result is always returned unchanged,
making this useful for side effects like logging or debugging errors.
If the result is :ok, the function is not called and the :ok result is returned unchanged.
Examples
# Monitor payment failures and send alerts
iex> process_payment(order)
...> |> tap_error(fn reason ->
...>   ErrorTracker.report(reason, context: %{order_id: order.id})
...>   if reason == :payment_gateway_down, do: PagerDuty.alert("Gateway down")
...> end)
{:error, :payment_gateway_down}

# Track metrics for failed operations
iex> update_inventory(product_id, quantity)
...> |> tap_error(fn reason ->
...>   ErrorTracker.report(reason, context: %{product_id: product_id})
...>   Metrics.increment("inventory.update.failed", tags: %{reason: reason})
...> end)
{:error, :insufficient_stock}

# Success results pass through without calling the function
iex> tap_error({:ok, user}, fn _ -> ErrorTracker.report("Won't be called") end)
{:ok, %User{}}

  



  
    
      
    
    
      tap_ok(result, func)



        
          
        

    

  


  

      

          @spec tap_ok(result(), (any() -> any())) :: result()


      


Executes a side-effect function when the result is :ok, passing the result through unchanged.
Takes a result and, if it's :ok or {:ok, value}, calls the provided function with
the unwrapped value (or nil for :ok). The result is always returned unchanged,
making this useful for side effects like logging or debugging.
If the result is an error, the function is not called and the error is returned unchanged.
Examples
# Track analytics for successful user registrations
iex> register_user(params)
...> |> tap_ok(fn user -> Analytics.track("user_registered", %{user_id: user.id}) end)
{:ok, %User{email: "user@example.com"}}

# Cache API responses while continuing the pipeline
iex> fetch_product_from_api(product_id)
...> |> tap_ok(fn product -> Cache.put("product:#{product_id}", product) end)
{:ok, %Product{id: 123, name: "Widget"}}

# Errors pass through without calling the function
iex> tap_ok({:error, :not_found}, fn user -> send_notification(user) end)
{:error, :not_found}

      all(error, func)



        
          
        

    

  


  

      

          @spec all(value_result() | Enumerable.t(), (any() -> result())) ::
  :ok | :error | {:error, Triage.WrappedError.t()}


      


Validates that a callback function gives an :ok / {:ok, _} result for all elements
in the given enumerable.
Takes an enumerable or {:ok, enumerable} (giving :ok will return an error)
and applies a callback function to each element. If all calls to the callback
return :ok or {:ok, any()} then all returns :ok.
If any callback returns an error, immediately stops processing and returns that error.
If :error or {:error, reason} are given as the first argument, they are returned
unchanged. Note that even if callbacks return {:ok, value}, the values are discarded
and only :ok is returned — this function is for validation, not transformation.
See map_if/2 if you need transformation which short-circuits.
This is useful when you need to validate that all items in a collection meet certain
criteria before proceeding with subsequent operations.
Examples
iex> Triage.all(emails, &check_valid_email)
:ok

iex> Triage.all({:ok, emails}, &check_valid_email)
{:error, :invalid_hostname}

iex> Triage.all(:error, fn _ -> <not called> end)
:error

iex> Triage.all({:error, :not_found}, fn _ -> <not called> end)
{:error, :not_found}

  



  
    
      
    
    
      find_value(error, func)



        
          
        

    

  


  

      

          @spec find_value(value_result() | Enumerable.t(), (any() -> result())) ::
  :ok | {:ok, any()} | {:error, [any()]}


      


Finds the first successful result from applying a function to enumerable elements.
Takes an enumerable or {:ok, enumerable} and applies a function to each element
The first successful result (:ok or {:ok, value}) from the callback is returned
from find_value and no further iteration is done.
If all callbacks return errors, then {:error, [list of error reasons]} is returned.
For :error atoms in the error list, they are represented as nil.
If :error or {:error, reason} is given as the first argument to find_value,
it is passed through unchanged.
This can be useful when you're trying multiple strategies or checking multiple
values to find the first one that works.
Examples
iex> Triage.find_value(domains, &ping_domain)
{:ok, "www.mydomain.com"}

iex> Triage.find_value({:ok, domains}, &ping_domain)
{:error, [:nxdomain, :timeout, :nxdomain]}

iex> Triage.find_value(:error, fn _ -> <not called> end)
:error

iex> Triage.find_value({:error, :not_found}, fn _ -> <not called> end)
{:error, :not_found}

  



  
    
      
    
    
      map_if(error, func)



        
          
        

    

  


  

      

          @spec map_if(result(), (any() -> any())) :: result()


      


Maps a function over an enumerable, collecting successful values and short-circuiting on the first error.
Takes an enumerable or {:ok, enumerable} and applies a function to each element.
If all callbacks return success with {:ok, value}), map_if returns
{:ok, [transformed_values]}. If any call to the callback returns an error, map_if
immediately stops processing and returns that error.
If map_if is given an :error result for it's first argument that argument is returned
unchanged and the callback is never called.
Since map_if requires an enumerable value to work with, it will fail if given :ok as a result argument.
This is useful when you need all transformations to succeed—if any fail, you don't want
the partial results.
Examples
iex> Triage.map_if(xml_docs, & xml_to_json(&1, opts))
{:ok, [...]}

iex> Triage.map_if(xml_docs, & xml_to_json(&1, opts))
{:error, ...}

iex> Triage.map_if(:error, fn _ -> <not called> end)
:error

iex> Triage.map_if({:error, :not_found}, fn _ -> <not called end)
{:error, :not_found}

      error?(result)



        
          
        

    

  


  

Checks if a result is an error (:error or {:error, any()}).
Returns true if the result is :error or {:error, any()}, false if it's
:ok or {:ok, any()}. Raises ArgumentError for any other value.
Examples
iex> Triage.error?(:error)
true

iex> Triage.error?({:error, :not_found})
true

iex> Triage.error?({:error, :not_found, 123})
true

iex> Triage.error?(:ok)
false

iex> Triage.error?({:ok, 42})
false

iex> Triage.error?({:ok, 42, :ignored})
false

  



    

  
    
      
    
    
      log(result, mode \\ :errors)



        
          
        

    

  


  

Logs a result tuple and returns it unchanged.
Takes a result and logs it. By default, only errors are logged.
The mode argument can be either :errors (the default) or :all (logs all results)
See this guide for information about
logging with JSON
The result passed in can be:
	:ok / :error
	{:ok, any()} / {:error, any()}
	{:ok, ...} / {:error, ...} (any sized tuple starting with :ok or :error)


  



  
    
      
    
    
      ok?(result)



        
          
        

    

  


  

Checks if a result is a success (:ok or {:ok, any()}).
Returns true if the result is :ok or {:ok, any()}, false if it's
:error or {:error, any()}. Raises ArgumentError for any other value.
Examples
iex> Triage.ok?(:ok)
true

iex> Triage.ok?({:ok, 42})
true

iex> Triage.ok?({:ok, 42, :ignore})
true

iex> Triage.ok?(:error)
false

iex> Triage.ok?({:error, :not_found})
false

iex> Triage.ok?({:error, :not_found, 123})
false

          @spec user_message({:error, any()}) :: String.t()


      


Generates a user-friendly error message from various error types.
Converts {:error, reason} tuples into human-readable messages suitable for displaying to end users.
When the reason is a string, the string error message is returned.
When the reason is Triage.WrappedError.t/0 it unwraps the error chain and includes context information in the message.
For exceptions and unknown error types, it
	generates a unique error code
	logs the error code with full error details
	returns a generic error to the user with the error code that the user can report

There is also logic to specifically deal with Ecto.Changeset validation errors
so that you don't need to implement your own (at least as long as you're happy
with the default behavior)
Parameters
	reason - The error to convert (string, exception, %Triage.WrappedError{}, or any other value)

Examples
iex> user_message({:error, "Invalid email"})
"Invalid email"

# WrappedError which was returned as a result of `wrap_context` being called in two places
# and where the original error was `{:error, "not found"}`
iex> user_message({:error, %Triage.WrappedError{}})
"not found (happened while: fetching user => validating email)"

# Ecto.Changeset with validation errors on multiple fields
iex> user_message({:error, %Ecto.Changeset{errors: [...]}})
"email: has invalid format, should be at least 10 character(s); name: can't be blank"

iex> user_message({:error, %RuntimeError{message: "boom"}})
"There was an error. Refer to code: ABC12345"

Log generated:
ABC12345: Could not generate user error message. Error was: #RuntimeError<...> (message: boom)

  


        

      


  

    
Triage.WrappedError exception
    



      
Exception struct which is returned in {:error, _} tuples when
the wrap_context module is used.
Contains
	context string
	context metadata
	an original error result (could be within nested WrappedErrors)

These things help make it clearer where an error is coming from.
Works well with the log and user_message functions.

          @type t() :: %Triage.WrappedError{
  __exception__: true,
  context: term(),
  message: term(),
  metadata: term(),
  result: term(),
  stacktrace: term()
}

